

NAG Library Routine Document

G05HNF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of **bold italicised** terms and other implementation-dependent details.

1 Purpose

G05HNF generates a given number of terms of an exponential GARCH(p, q) process (see Engle and Ng (1993)).

2 Specification

```

SUBROUTINE G05HNF(DIST, NUM, IP, IQ, THETA, DF, HT, ET, FCALL, RVEC,
1              IGEN, ISEED, RWSAV, IFAIL)
      INTEGER          NUM, IP, IQ, IGEN, ISEED(4), IFAIL
      double precision THETA(2*IQ+IP+1), DF, HT(NUM), ET(NUM), RVEC(40),
1              RWSAV(9)
      LOGICAL          FCALL
      CHARACTER*1      DIST

```

3 Description

An exponential GARCH(p, q) process is represented by:

$$\ln(h_t) = \alpha_0 + \sum_{i=1}^q \alpha_i z_{t-i} + \sum_{i=1}^q \phi_i (|z_{t-i}| - E[|z_{t-i}|]) + \sum_{j=1}^p \beta_j \ln(h_{t-j}), \quad t = 1, 2, \dots, T,$$

where $z_t = \frac{\epsilon_t}{\sqrt{h_t}}$, $E[|z_{t-i}|]$ denotes the expected value of $|z_{t-i}|$, and $\epsilon_t | \psi_{t-1} = N(0, h_t)$ or $\epsilon_t | \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t -distribution with df degrees of freedom and variance h_t , T is the number of observations in the sequence, ϵ_t is the *observed* value of the GARCH(p, q) process at time t , h_t is the conditional variance at time t , and ψ_t the set of all information up to time t .

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05HNF.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and Testing the Impact of News on Volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

- 1: DIST – CHARACTER*1 *Input*
On entry: the type of distribution to use for ϵ_t .
 DIST = 'N'
 A Normal distribution is used.
 DIST = 'T'
 A Student's t -distribution is used.
Constraint: DIST = 'N' or 'T'.
- 2: NUM – INTEGER *Input*
On entry: T , the number of terms in the sequence.
Constraint: NUM > 0.
- 3: IP – INTEGER *Input*
On entry: the number of coefficients, β_i , for $i = 1, 2, \dots, p$.
Constraints:
 $2 \times IQ + IP + 1 \leq 20$;
 $IP \geq 0$.
- 4: IQ – INTEGER *Input*
On entry: the number of coefficients, α_i , for $i = 1, 2, \dots, q$.
Constraints:
 $2 \times IQ + IP + 1 \leq 20$;
 $IQ \geq 1$.
- 5: THETA(2 × IQ + IP + 1) – *double precision* array *Input*
On entry: the initial parameter estimates for the vector θ . The first element must contain the coefficient α_0 and the next IQ elements must contain the autoregressive coefficients α_i , for $i = 1, 2, \dots, q$. The next IQ elements must contain the coefficients ϕ_i , for $i = 1, 2, \dots, q$. The next IP elements must contain the moving average coefficients β_j , for $j = 1, 2, \dots, p$.
- 6: DF – *double precision* *Input*
On entry: the number of degrees of freedom for the Student's t -distribution.
 If DIST = 'N', DF is not referenced.
Constraint: if DIST = 'T', DF > 2.
- 7: HT(NUM) – *double precision* array *Output*
On exit: the conditional variances h_t , for $t = 1, 2, \dots, T$, for the GARCH(p, q) sequence.
- 8: ET(NUM) – *double precision* array *Output*
On exit: the observations ϵ_t , for $t = 1, 2, \dots, T$, for the GARCH(p, q) sequence.
- 9: FCALL – LOGICAL *Input*
On entry: if FCALL = .TRUE., a new sequence is to be generated, otherwise a given sequence is to be continued using the information in RVEC.
- 10: RVEC(40) – *double precision* array *Input/Output*
On entry: the array contains information required to continue a sequence if FCALL = .FALSE..

On exit: contains information that can be used in a subsequent call of G05HNF, with FCALL = .FALSE..

11: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

12: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

13: RWSAV(9) – *double precision* array *Communication Array*

14: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or –1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, IP < 0,
or IQ < 1,
or DF ≤ 2,
or NUM ≤ 0,
or DIST ≠ 'N' and DIST ≠ 'T',
or $2 \times IQ + IP + 1 > 20$.

IFAIL = 2

Invalid sequence generated, use different parameters.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example uses G05HNF to generate 1500 data points, with known process parameters θ for the following two time series:

- (i) a GARCH(1,1) sequence with Normally distributed residuals;
- (ii) a GARCH(1,2) sequence with Student's t -distributed residuals.

Here G05HNF is initially called, with the output discarded, to eliminate 'start-up effects' in these sequences. The process parameter estimates, $\hat{\theta}$, are then obtained using G13FGF, and compared with their true values, θ . Finally a four step ahead volatility estimate is computed using G13FHF.

9.1 Program Text

```

*      G05HNF Example Program Text
*      Mark 22 Release. NAG Copyright 2008.
*      .. Parameters ..
INTEGER          NOUT
PARAMETER        (NOUT=6)
INTEGER          NPARMX, LDICOVR, NUM
DOUBLE PRECISION ZERO
PARAMETER        (NPARMX=10,LDICOVR=NPARMX,NUM=1500,ZERO=0.0D0)
INTEGER          NUM1, NREGMX
PARAMETER        (NUM1=3000,NREGMX=10)
*      .. Local Scalars ..
DOUBLE PRECISION DF, FAC1, HP, LGF, MEAN, TOL, XTERM
INTEGER          I, IFAIL, IGEN, IP, IQ, K, LDX, LWORK, MAXIT, MN,
+               NPAR, NPAR2, NREG, NT
LOGICAL          COPT, FCALL
CHARACTER        DIST
*      .. Local Arrays ..
DOUBLE PRECISION BX(10), COVR(LDICOVR,NPARMX), CVAR(100),
+               ETM(NUM1), HT(NUM1+10), HTM(NUM1), PARAM(NPARMX),
+               RVEC(40), RWSAV(9), SC(NPARMX), SE(NPARMX),
+               THETA(NPARMX), WORK(NUM1*3+NPARMX+NREGMX*NUM1+20*
+               20+1), X(NUM1,10), YT(NUM1+10)
INTEGER          ISEED(4)
*      .. External Subroutines ..
EXTERNAL         G05HNF, G05KBF, G13FGF, G13FHF
*      .. Intrinsic Functions ..
INTRINSIC        DBLE, SIN
*      .. Executable Statements ..
WRITE (NOUT,*) 'G05HNF Example Program Results'
WRITE (NOUT,*)

*
*      Initialize the generator
ISEED(1) = 111
IGEN = 0
CALL G05KBF(IGEN,ISEED)
*
*      Example using Normal errors
LDX = NUM1
BX(1) = 1.5D0
BX(2) = 2.5D0
MEAN = 3.0D0
*
DO 20 I = 1, NUM
    FAC1 = DBLE(I)*0.01D0
    X(I,1) = 0.01D0 + 0.7D0*SIN(FAC1)
    X(I,2) = 0.5D0 + FAC1*0.1D0
    X(I,3) = 1.0D0
20 CONTINUE
*
NREG = 2
MN = 1
IP = 1
IQ = 1
NPAR = IP + 2*IQ + 1
*
PARAM(1) = 0.1D0
PARAM(2) = -0.3D0
PARAM(3) = 0.1D0
PARAM(4) = 0.9D0
*

```

```

      DF = 5.0D0
      DIST = 'N'
*
*   Call the routine for the first time
      FCALL = .TRUE.
      IFAIL = 0
      CALL G05HNF(DIST,800,IP,IQ,PARAM,DF,HT,YT,FCALL,RVEC,IGEN,ISEED,
+               RWSAV,IFAIL)
*   Discard the first 800 values, and call the routine again
      FCALL = .FALSE.
      IFAIL = 0
      CALL G05HNF(DIST,NUM,IP,IQ,PARAM,DF,HT,YT,FCALL,RVEC,IGEN,ISEED,
+               RWSAV,IFAIL)
*
      DO 60 I = 1, NUM
        XTERM = ZERO
        DO 40 K = 1, NREG
          XTERM = XTERM + X(I,K)*BX(K)
40      CONTINUE
        IF (MN.EQ.1) THEN
          YT(I) = MEAN + XTERM + YT(I)
        ELSE
          YT(I) = XTERM + YT(I)
        END IF
      60 CONTINUE
*
      COPT = .TRUE.
*
      MAXIT = 50
      TOL = 1.0D-5
*
      DO 80 I = 1, NPAR
        THETA(I) = PARAM(I)*0.5D0
80      CONTINUE
*
      IF (MN.EQ.1) THEN
        THETA(NPAR+MN) = MEAN*0.5D0
      END IF
*
      DO 100 I = 1, NREG
        THETA(NPAR+MN+I) = BX(I)*0.5D0
100     CONTINUE
*
*   Fit an EGARCH model to the generated data
      LWORK = NREG*NUM + 3*NUM + 3
      NPAR2 = 1 + IQ*2 + IP + MN + NREG
      IFAIL = 0
      CALL G13FGF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,NPAR2,THETA,SE,SC,
+               COVR,LDCOVR,HP,ETM,HTM,LGF,COPT,MAXIT,TOL,WORK,LWORK,
+               IFAIL)
*   Display the output
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Normal distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '          Correct      Parameter      Standard'
      WRITE (NOUT,*) '          values      estimates      errors'
      DO 120 I = 1, NPAR
        WRITE (NOUT,99998) 'PARAM(', I, ')', PARAM(I), THETA(I), SE(I)
120     CONTINUE
      IF (MN.EQ.1) THEN
        WRITE (NOUT,99999) 'Mean      ', MEAN, THETA(NPAR+1), SE(NPAR+1)
      END IF
      DO 140 I = 1, NREG
        WRITE (NOUT,99998) 'BX(', I, ')      ', BX(I), THETA(NPAR+MN+I),
+               SE(NPAR+MN+I)
140     CONTINUE
*
      NT = 4
      IFAIL = 0
      CALL G13FHF(NUM,NT,IP,IQ,THETA,CVAR,HTM,ETM,IFAIL)
      WRITE (NOUT,*)

```

```

WRITE (NOUT,99997) 'Volatility forecast = ', CVAR(NT)
WRITE (NOUT,*)
*
* Example using Students T errors
NREG = 2
MN = 1
IP = 1
IQ = 2
NPAR = IP + 2*IQ + 1
*
PARAM(1) = 0.1D0
PARAM(2) = -0.3D0
PARAM(3) = -0.1D0
PARAM(4) = 0.1D0
PARAM(5) = 0.3D0
PARAM(6) = 0.7D0
*
DIST = 'T'
*
* Call the routine the first time
FCALL = .TRUE.
IFAIL = 0
CALL G05HNF(DIST,NUM,IP,IQ,PARAM,DF,HT,YT,FCALL,RVEC,IGEN,ISEED,
+          RWSAV,IFAIL)
* Call the routine the a second time
FCALL = .FALSE.
IFAIL = 0
CALL G05HNF(DIST,NUM,IP,IQ,PARAM,DF,HT,YT,FCALL,RVEC,IGEN,ISEED,
+          RWSAV,IFAIL)
* Discard all previously generated values, and call the routine again
FCALL = .FALSE.
IFAIL = 0
CALL G05HNF(DIST,NUM,IP,IQ,PARAM,DF,HT,YT,FCALL,RVEC,IGEN,ISEED,
+          RWSAV,IFAIL)
*
DO 180 I = 1, NUM
  XTERM = ZERO
  DO 160 K = 1, NREG
    XTERM = XTERM + X(I,K)*BX(K)
160  CONTINUE
  IF (MN.EQ.1) THEN
    YT(I) = MEAN + XTERM + YT(I)
  ELSE
    YT(I) = XTERM + YT(I)
  END IF
180 CONTINUE
*
COPT = .TRUE.
MAXIT = 50
TOL = 1.0D-5
*
DO 200 I = 1, NPAR
  THETA(I) = PARAM(I)*0.5D0
200 CONTINUE
*
THETA(NPAR+1) = DF*0.65D0
*
IF (MN.EQ.1) THEN
  THETA(NPAR+1+MN) = MEAN*0.5D0
END IF
*
DO 220 I = 1, NREG
  THETA(NPAR+1+MN+I) = BX(I)*0.5D0
220 CONTINUE
*
* Fit an EGARCH model to the generated data
LWORK = NREG*NUM + 3*NUM + 3
NPAR2 = 2 + IQ*2 + IP + MN + NREG
IFAIL = 0
CALL G13FGF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,NPAR2,THETA,SE,SC,
+          COVR,LDCOVR,HP,ETM,HTM,LGF,COPT,MAXIT,TOL,WORK,LWORK,

```

```

+           IFAIL)
*
WRITE (NOUT,*)
WRITE (NOUT,*) 'Students t distribution'
WRITE (NOUT,*)
WRITE (NOUT,*) '          Correct      Parameter      Standard'
WRITE (NOUT,*) '          values      estimates      errors'
DO 240 I = 1, NPAR
  WRITE (NOUT,99998) 'PARAM(', I, ')', PARAM(I), THETA(I), SE(I)
240 CONTINUE
  WRITE (NOUT,99999) 'DF          ', DF, THETA(NPAR+1), SE(NPAR+1)
  IF (MN.EQ.1) THEN
    WRITE (NOUT,99999) 'Mean      ', MEAN, THETA(NPAR+1+MN),
+      SE(NPAR+1+MN)
  END IF
  DO 260 I = 1, NREG
    WRITE (NOUT,99998) 'BX(', I, ')      ', BX(I), THETA(NPAR+1+MN+I),
+      SE(NPAR+1+MN+I)
  260 CONTINUE
*
NT = 4
IFAIL = 0
CALL G13FHF(NUM,NT,IP,IQ,THETA,CVAR,HTM,ETM,IFAIL)
WRITE (NOUT,*)
WRITE (NOUT,99997) 'Volatility forecast = ', CVAR(NT)
WRITE (NOUT,*)
*
99999 FORMAT (1X,A,3(F10.2,2X))
99998 FORMAT (1X,A,I1,A,3(F10.2,2X))
99997 FORMAT (1X,A,F12.2)
END

```

9.2 Program Data

None.

9.3 Program Results

G05HNF Example Program Results

Normal distribution

	Correct values	Parameter estimates	Standard errors
PARAM(1)	0.10	0.12	0.02
PARAM(2)	-0.30	-0.31	0.03
PARAM(3)	0.10	0.12	0.03
PARAM(4)	0.90	0.89	0.02
Mean	3.00	2.86	0.08
BX(1)	1.50	1.45	0.04
BX(2)	2.50	2.58	0.06

Volatility forecast = 2.95

Students t distribution

	Correct values	Parameter estimates	Standard errors
PARAM(1)	0.10	0.09	0.03
PARAM(2)	-0.30	-0.36	0.05
PARAM(3)	-0.10	-0.13	0.06
PARAM(4)	0.10	0.21	0.07
PARAM(5)	0.30	0.24	0.08
PARAM(6)	0.70	0.69	0.04
DF	5.00	5.60	0.71
Mean	3.00	2.93	0.07

BX(1)	1.50	1.45	0.04
BX(2)	2.50	2.56	0.05

Volatility forecast = 0.99
